> Soluzioni del compito di Calcolo Numerico del 7 gennaio 2004

> Enrico Bertolazzi

```
> restart:
  with(LinearAlgebra) :
```

Esercizio 1

Funzione di partenza

```
> f := x -> x - exp(-x) ;
```

$$f := x \to x - e^{(-x)}$$

Derivata prima

```
> df := D(f);
```

$$df := x \to 1 + e^{(-x)}$$

Metodo di Newton

```
> Newton := unapply( factor(x - f(x) / df(x)), x);
```

$$Newton := x \to \frac{e^{(-x)} (x+1)}{1 + e^{(-x)}}$$

Metodo delle secanti

```
> Secanti := unapply( factor( (y*f(x)-x*f(y)) / (f(x)-f(y))), x, y);
```

$$Secanti := (x, y) \to \frac{-y\, e^{(-x)} + x\, e^{(-y)}}{x - e^{(-x)} - y + e^{(-y)}}$$

Metodo di Newton a partire da x0 = 1:

```
> x0 := 1:
  x1 := evalf(Newton(x0),4):
  x2 := evalf(Newton(x1),4):
  x3 := evalf(Newton(x2),4):
  x4 := evalf(Newton(x3),4):
  <x0,x1,x2,x3,x4>;
```

$$\begin{bmatrix} 1 \\ 0.5378 \\ 0.5670 \\ 0.5672 \\ 0.5671 \end{bmatrix}$$

Metodo delle secanti a partire da x0 = 0 x1 = 1 ;

```
> x0 := 0 :
  x1 := 2 :
  x2 := evalf(Secanti(x0,x1),4):
  x3 := evalf(Secanti(x1,x2),4):
  x4 := evalf(Secanti(x2,x3),4):
  <x0,x1,x2,x3,x4>;
```

$$\begin{bmatrix} 0 \\ 2 \\ 0.6980 \\ 0.5410 \\ 0.5679 \end{bmatrix}$$

Esercizio 2

Punti di interpolazione:

```
> X := [-1,1,-2,2,-3];
  Y := [1,1,2,2,7];
```

$$X := [-1, 1, -2, 2, -3]$$

$$Y := [1, 1, 2, 2, 7]$$

Polinomio interpolante:

```
> interp( X, Y, 'z');
```

$$\frac{1}{12} z^4 - \frac{1}{12} z^2 + 1$$

Costruzione delle differenze divise di ordine 0

```
> f1 := Y[1];
  f2 := Y[2];
  f3 := Y[3];
  f4 := Y[4];
  f5 := Y[5];
```

$$f1 := 1$$

$$f2 := 1$$

$$f3 := 2$$

$$f4 := 2$$

$$f5 := 7$$

Differenze divise

```
> f12 := (f2-f1)/(X[2]-X[1]);
  f23 := (f3-f2)/(X[3]-X[2]);
  f34 := (f4-f3)/(X[4]-X[3]);
  f45 := (f5-f4)/(X[5]-X[4]);
```

$$f12 := 0$$

$$f23 := \frac{-1}{3}$$

$$f34 := 0$$

$$f45 := -1$$

Differenze divise seconde

```
> f123 := (f23-f12)/(X[3]-X[1]);
  f234 := (f34-f23)/(X[4]-X[2]);
  f345 := (f45-f34)/(X[5]-X[3]);
```

$$f123 := \frac{1}{3}$$

$$f234 := \frac{1}{3}$$

$$f345 := 1$$

Differenze divise terze

```
> f1234 := (f234-f123)/(X[4]-X[1]);
  f2345 := (f345-f234)/(X[5]-X[2]);
```

$$f1234 := 0$$

$$f2345 := \frac{-1}{6}$$

Differenze divise quarte

```
> f12345 := (f2345-f1234)/(X[5]-X[1]);
```

$$f12345 := \frac{1}{12}$$

Polinomi della base

```
> w0 := 1 ;
  w1 := x-X[1] ;
  w2 := expand(w1 * ( x - X[2])) ;
  w3 := expand(w2 * ( x - X[3])) ;
  w4 := expand(w3 * ( x - X[4])) ;
```

$$w0 := 1$$

$$w1 := x + 1$$

$$w2 := x^2 - 1$$

$$w3 := x^3 + 2\,x^2 - x - 2$$

$$w4 := x^4 - 5 x^2 + 4$$

Polinomio interpolante

```
> p := f1*w0 + f12 * w1 + f123 * w2 + f1234 * w3 + f12345 * w4 ;
```

$$p := 1 - \frac{1}{12} x^2 + \frac{1}{12} x^4$$

Esercizio 3

Funzione da integrare:

```
> f := x -> sin(3*x)-x*cos(x) ;
```

$$f := x \rightarrow \sin(3\ x) - x \cos(x)$$

Integrale esatto

```
> a,b := -2,1:
  int(f(x),x=a..b);
  esatto := evalf(%);
```

$$\frac{1}{3} \cos(6) + \cos(2) + 2 \sin(2) - \frac{1}{3} \cos(3) - \cos(1) - \sin(1)$$

$$esatto := 0.6707289882$$

Derivata seconda della funzione

```
> ddf := (D@D)(f) ;
```

$$ddf := x \rightarrow -9 \sin(3\ x) + 2 \sin(x) + x \cos(x)$$

Possibile costante che maggiore il modulo della derivata seconda

```
> M := 9 + 2 + 2 ;
```

$$M := 13$$

Verifica grafica:

```
> plot(abs(ddf(x)),x=a..b) ;
```



Calcolo in numero minimo stimato di intervalli

```
> err := 10^(-3) = (b-a)^3/(12*n^2)*M ;
```

$$err := \frac{1}{1000} = \frac{117}{4\ n^2}$$

```
> isolate(err,n^2) ;
```

$$n^2 = 29250$$

```
> evalf(sqrt(rhs(%)));
```

$$171.0263138$$

Intervalli stimati: 172

```
> h := (b-a)/172 ;
```

$$h := \frac{3}{172}$$

```
> trap := evalf(h*(f(a)+f(b))/2+h*sum(f(a+h*i),i=1..171)) ;
```

$$trap := 0.6705316417$$

```
> trap - esatto ;
```

$$-0.0001973465$$

Integrale con 6 intervalli

```
> h := (b-a)/6 ;
```

$$h := \frac{1}{2}$$

```
> trap := evalf(h*(f(a)+f(b))/2+h*sum(f(a+h*i),i=1..5)) ;
```

$$trap := 0.5033939910$$

Esercizio 4

```
> sol := x -> 1-x^2;
```

$$sol := x \rightarrow 1 - x^2$$

Definizione del problema

```
> p := x -> x ;
  q := x -> -2 ;
  r := unapply(diff(sol(x),x,x)+p(x)*diff(sol(x),x)+q(x)*sol(x),x) ;
  xa, xb := -2, 2 ;
  ya, yb := sol(xa), sol(xb) ;
```

$$p := x \rightarrow x$$

$$q := x \rightarrow -2$$

$$r := x \rightarrow -4$$

$$xa, xb := -2, 2$$

$$ya, yb := -3, -3$$

Differenze finite:

```
> n := 4 ;
  h := (xb-xa) / n ;
  x[0] := xa ;
  x[1] := xa + h ;
  x[2] := xa + 2*h ;
  x[3] := xa + 3*h ;
  x[4] := xa + 4*h ;
```

$$n := 4$$

$$h := 1$$

$$x_0 := -2$$

$$x_1 := -1$$

$$x_2 := 0$$

$$x_3 := 1$$

$$x_4 := 2$$

```
> eq := k -> (y[k+1]-2*y[k]+y[k-1])/h^2 +
             p(x[k]) * (y[k+1]-y[k-1])/(2*h) +
             q(x[k]) * y[k] - r(x[k]) ;
```

$$eq := k \rightarrow \frac{y_{k+1} - 2\, y_k + y_{k-1}}{h^2} + \frac{p(x_k)\, (y_{k+1} - y_{k-1})}{2\, h} + q(x_k)\, y_k - r(x_k)$$

Equazioni risultanti

```
> eq1 := eq(1) ;
  eq2 := eq(2) ;
  eq3 := eq(3) ;
```

$$eq1 := \frac{1}{2} y_2 - 4\, y_1 + \frac{3}{2} y_0 + 4$$

$$eq2 := y_3 - 4\, y_2 + y_1 + 4$$

$$eq3 := \frac{3}{2} y_4 - 4\, y_3 + \frac{1}{2} y_2 + 4$$

Estraggo il sistema lineare dalle equazioni

```
> A := linalg[genmatrix]([eq1,eq2,eq3],[y[1],y[2],y[3]],'b'):
  A := convert(A,Matrix);
```

$$A := \begin{bmatrix} -4 & \frac{1}{2} & 0 \\ 1 & -4 & 1 \\ 0 & \frac{1}{2} & -4 \end{bmatrix}$$

```
> b := Transpose(convert(b,Vector));
```

$$b := \begin{bmatrix} -\frac{3}{2} y_0 - 4 \\ -4 \\ -\frac{3}{2} y_4 - 4 \end{bmatrix}$$

Sostituisco le condizioni al contorno

```
> b := subs(y[0]=ya, y[4]=yb, b) ;
```

$$b := \begin{bmatrix} \frac{1}{2} \\ -4 \\ \frac{1}{2} \end{bmatrix}$$

Risolvo il sistema lineare

```
> LinearSolve(A,b) ;
```

$$\begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}$$

Esercizio 5

Tabella dei punti

```
> X := [-2,-2,-1,0,1,2,2,3,2,1];
  Y := [-1,-1,0,1,3,2,4,-2,3,1];
```

$$X := [-2, -2, -1, 0, 1, 2, 2, 3, 2, 1]$$

$$Y := [-1, -1, 0, 1, 3, 2, 4, -2, 3, 1]$$

```
> n := nops(X) ;
```

$$n := 10$$

```
> SX   := add(X[i],i=1..n) ;
  SX2  := add(X[i]^2,i=1..n) ;
  SX3  := add(X[i]^3,i=1..n) ;
  SX4  := add(X[i]^4,i=1..n) ;
  SY   := add(Y[i],i=1..n) ;
  SXY  := add(X[i]*Y[i],i=1..n) ;
  SX2Y := add(X[i]^2*Y[i],i=1..n) ;
```

$$SX := 6$$

$$SX2 := 32$$

$$SX3 := 36$$

$$SX4 := 164$$

$$SY := 10$$

$$SXY := 20$$

$$SX2Y := 14$$

```
> A := <<n,SX,SX2>|<SX,SX2,SX3>|<SX2,SX3,SX4>>;
```

$$A := \begin{bmatrix} 10 & 6 & 32 \\ 6 & 32 & 36 \\ 32 & 36 & 164 \end{bmatrix}$$

```
> b := <SY,SXY,SX2Y> ;
```

$$b := \begin{bmatrix} 10 \\ 20 \\ 14 \end{bmatrix}$$

```
> res := A^(-1).b ;
```

$$res := \begin{bmatrix} \frac{1973}{917} \\ \frac{104}{131} \\ \frac{-933}{1834} \end{bmatrix}$$

Soluzione

```
> x := 'x' :
  p := 'p' :
```

```
> p := res[1]+ x * res[2] + x^2 * res[3] ;
```

$$p := \frac{1973}{917} + \frac{104}{131} x - \frac{933}{1834} x^2$$

```
> evalf(p,4) ;
```

$$2.152 + 0.7939 x - 0.5087 x^2$$

```
>
```